PARI/GP

L-FUNCTIONS

Marine Rougnant

Université de Franche-Comté (Besançon, France)

Institute of Mathematical Sciences (Chennai, India)
19/02/24 – 23/02/24

L-function constructors

To define a L-function, use `lfuncreate`. The entry will determine which L-function you are defining :

Riemann ζ	`1`
Dirichlet for quadratic char. $(D/\cdot)$	`D`
Dirichlet L-series	`Mod(m,N)`
Dedekind ζ_K, $K = \mathbb{Q}[X]/(T)$	`T`
Hecke L-series, χ mod $\mathfrak{f}$	`[bnr,` *chi*`]`
Artin L-function	`lfunartin` (/!\\)
$L(E,s)$, E elliptic curve	`E`
genus 2 curve, $y^2 = F(x)$	`lfungenus2(F)` (/!\\)
..., $y^2 + Q(x)y = F(x)$	`lfungenus2([F,Q])` (/!\\)

Then initialize $L(s)$ with `lfuninit(L, domain)` where :

- L is the output of one of the functions `lfuncreate`, `lfunartin`, `lfungenus2`
- *domain* is a rectangle centered on the real axis and defined by its center, width and height `[c,w,h]` :
  $|\Re(s) - c| \leqslant w$, and $|\Im(s)| \leqslant h$
  The subdomain `[k/2, 0, h]` on the critical line can be encoded as `[h]`.
  If you only need $L(s)$ for real s, set h to 0.

`lfuncreate` for $L(s) = \sum_{n \geqslant 1} a_n(s) n^{-s}$ output contains :

- Dirichlet coefficients $a(n)$,
- Dirichlet coefficients $a^*(n)$ for dual L-function L^*,
- Euler gamma factor $A = [a_1, \ldots, a_d]$ for $\gamma_A(s) = \prod_i \Gamma_{\mathbb{R}}(s + a_i)$,
- Weight k (valuea at s and $k - s$ related by the functional equation)
- Conductor N, $\Lambda(s) = N^{s/2}\gamma_A(s)$
- Root number ε, $\Lambda(a, k - s) = \varepsilon\Lambda(a^*, s)$
- Polar part

$$\zeta(s) = \sum_{n\geqslant 1} n^{-s}$$

```
? Zeta = lfuncreate(1) \\ encodes the Riemann zeta function
% = [[Vecsmall([1]),1],0,[0],1,1,1,1]
? lfunan(L,20)  \\first 20 coefficient of the Dirichlet series
? lfun(Zeta,2) \\ L(2)
% = 1.6449340668482264364724151666460251892
? lfun(Zeta,0,1) \\ derivative of order 1 at 0
% = -0.91893853320467274178032973640561763986
? lfun(Zeta,1)
% = 1.00000000000*x^-1+O(x^0)
? lfun(Zeta,1+x+O(x^10))
% = 1.000000000*x^-1+0.5772156+0.0728158*x-0.004845*x^2-...
? lfunzeros(Zeta,20) \\zeros up to heigt 20
% = [14.134725141734693790457251983562470271]
? lfunlambda(Zeta,2) \\completed L-function Lambda(2)
% = 0.52359877559829887307710723054658381403
```

$$L(s,\chi) = \sum_{n\geqslant 1} \chi(n) n^{-s}$$

```
? G=znstar(4,1); G.clgp \\structure of  (Z/nZ)^*
% = [2, [2], [3]]
? Dir=lfuncreate([G,[1]]); Dir[2..5]
% = [0,[1],1,4]
```

$\chi = \left(\frac{1}{\cdot}\right)$, (Kronecker character, primitive)
The dual fonction has the same coefficients as $\beta(s)$
Only one term in the gamma factor's product ($\alpha_1 = 1$) :

$$\gamma_A(s) = \pi^{-(s+1)/2}\Gamma\left(\frac{s+1}{2}\right)$$

The functional equation relates values at s and $1-s$.
The conductor is $N = 4$: $\Lambda(s) = 4^{\frac{s}{2}}\gamma_A(s)L(s)$

```
? lfunan(Dir,30) \\first coeff. of the Dirichlet series
% = [1,0,-1,0,1,0,-1,0,1,0,-1,0,1,0,-1,0,1,0,-1,...]
```

$$L(s,\chi) = \sum_{n\geqslant 0,} (-1)^n (2n+1)^{-s}$$

```
? lfun(Dir,2) \\ L(2), Catalan's constant
% = 0.91596559417721901505460351493238411078
? Catalan
% = 0.91596559417721901505460351493238411077
```

$$\zeta_K(s) = \sum_{I \subseteq \mathcal{O}_K} \frac{1}{(N_{K/\mathbb{Q}}(I))^s}$$

```
? Dedek = lfuncreate(x^2+1); Dedek[2..5] \\K=Q(i)
%14 = [0,[0,1],1,4]
```

The dual fonction has the same coefficients than $\beta(s)$
Two terms in the gamma factor's product $(\alpha_1 = 0, \alpha_2 = 1)$:

$$\gamma_A(s) = \pi^{-s/2}\pi^{-(s+1)/2}\Gamma\left(\frac{s}{2}\right)\Gamma\left(\frac{s+1}{2}\right)$$

The functional equation relates values at s and $1-s$.
The conductor is $N = 4$: $\Lambda(s) = 4^{\frac{s}{2}}\gamma_A(s)L(s)$

```
? lfun(Dedek,2)
%15 = 1.5067030099229850308865650481820713960
? zeta(2)*Catalan
%16 = 1.5067030099229850308865650481820713960
? L=lfunmul(Zeta,Mod(3,4)); \\product of Dirichlet series
? lfun(L,2)
%18 = 1.5067030099229850308865650481820713960
? L2=lfundiv(Dedek,1);\\quotient of Dirichlet series
? lfun(L2,2)
%20 = 0.91596559417721901505460351493238411078
```

We define the elliptic curve $y^2 + 1 = x^3 + x^2 - 7x + 6$ over $\mathbb{Q}$.

```
? E = ellinit([0,0,1,-7,6]); L = lfuncreate(E); \\ L(E, s)
? lfun(L, 1)
% = 0.E-58
? lfun(E, 1, 1) \\ L'(1) = 0
% = 1.0280697501645834273549120167678691687 E-41
? lfun(E, 1, 2) \\ L(2)(1) = 0
% = 2.6769182259726016729068463995455020017 E-41
? lfun(E, 1, 3) \\ L(3)(1) != 0 , zero of order 3
% = 10.391099400715804138751850510360917070
? ellanalyticrank(E)
% = [3, 10.391099400715804138751850510360917070]
? lfunzeros(E,10)
% = [0, 0, 0, 2.052..., 3.262..., 4.470..., 4.754..., ...
```

The rank of $E(K)$ is the order of the zero of $L(E, s)$ at $s = 1$: BSD checked !

Twist by a Dirichlet character

`lfuntwist` allows to twist an L function by a Dirichlet character. The conductors need to be coprime.

```
? E = ellinit([0,-1,1,-10,-20]);
? L=lfuntwist(E,Mod(2,5));
? lfunan(E,10) \\ Dirichlet series coeff for L(E,s)
%3 = [1,-2,-1,2,1,2,-2,0,-2,-2]
? lfunan(Mod(2,5),10) \\Dirichlet character
%4 = [1,I,-I,-1,0,1,I,-I,-1,0]
? lfunan(L,10) \\Dirichlet series coeff for the twist
%5 = [1,-2*I,I,-2,0,2,-2*I,0,2,0]
```

We redefine the curve over $\mathbb{Q}(\zeta_5)$.

```
? nf=nfinit(polcyclo(5,'a));
? E2=ellinit(E[1..5],nf);
? localbitprec(64); lfun(E2,2)
%8 = 1.0543811873412420765
? L2=lfuntwist(E,Mod(4,5));
? lfun(E,2)*lfun(L2,2)*norm(lfun(L,2))
%10 = 1.0543811873410821651289745964738865962
```

For the genus-2 curve $y^2 + (x^3 + 1)y = x^2 + x$:

```
? L=lfungenus2([x^2+x,x^3+1]);
? L[2..5]
%12 = [0,[0,0,1,1],2,249]
? lfun(L,1)
%13 = 0.13154950701147875921340134301217526069
? lfunan(L,5)
%14 = [1,-2,-2,1,0]
```

$$L(s, A) = 1 - \frac{2}{2^s} - \frac{2}{3^s} + \frac{1}{4^s} + \dots$$

```
? bnf = bnfinit(a^2+23); \\initialize the number field Q(sqrt(-23
? bnr = bnrinit(bnf, 1); \\ ray class group for mdulus 1
? bnr.clgp
%3 = [3,[3]]
? Hecke = lfuncreate([bnr,[1]]);
? Hecke[2..5]
? z=lfun(Hecke,0,1) \\ L'(0)
%4 = 0.28119957432296184651205076406787829979+0.E-60*I
? algdep(exp(z),3) \\pol having exp(z) as approximate root
%5 = x^3-x-1
```

We start with a Galois extension of the rationals, here $\mathbb{Q}(\sqrt[3]{2}, \zeta_3) = \mathbb{Q}(\sqrt[6]{-108})$, with Galois group isomorphic to $S3$.

```
? N = nfinit(x^6+108);
? G = galoisinit(N);
```

G is the Galois group of N.

```
? [T,o] = galoischartable(G);
? T~
%4 = [1 1 1]
% [1 1 -1]
% [2 -1 0]
```

T is the character table of $G \cong S3$, which is defined over $\mathbb{Z}$. The first character is related to the trivial representation, the second to the signature, and the third to a faithful irreducible representation of dimension 2. The ordering of the conjugacy classes is given by `galoisconjclasses(G)`.

```
? galoisconjclasses(G)
%4 = [[Vecsmall([1,2,3,4,5,6])],[Vecsmall([3,1,2,6,4,5]),Vecsmall
```

We compute the Artin L-function associated to the 3rd character.

```
? L = lfunartin(N,G,T[,3],o);
? lfuncheckfeq(L)
%6 = -127
? L[2..5]
%7 = [0, [0, 1], 1, 108]
? z = lfun(L,0,1)
%8 = 1.3473773483293841009181878914456530463
? algdep(exp(z),3)
%9 = x^3-3*x^2-3*x-1
```

which suggests that this function is equal to a Hecke L-function.

```
? bnr = bnrinit(bnfinit(a^2+a+1),6);
? lfunan([bnr,[1]],100)==lfunan(L,100)
%11 = 1
```